
ABSTRACT  
Laboratory toxicity grading has been an important part of safety reporting since the FDA began accepting 
electronic data in 1999. Staying up to date with the various CTCAE versions can be a challenge. CTCAE 
Version 5.0 adds a layer of complexity with new grading criteria dependent on baseline measurements. 
We will present a practical method for deriving toxicity grades in the SDTM LB domain based on the new 
CTCAE, as well as reporting toxicity events in an OCCDS dataset derived separately from the BDS 
dataset with the laboratory findings. 

INTRODUCTION 
Toxicity grading for laboratory results began in 1999 with CTCAE version 2.0.  In this version, toxicity 
grading was based on absolute ranges relative to the upper and lower limits of normal.  With these 
simplistic criteria, deriving the toxicity grades was a simple task.  However, the grading criteria became 
more complex with the release of CTCAE version 4.0 in 2009 quickly replaced by version 4.03 in 2010.  
In this version, the change in the laboratory measurement from baseline also became a factor in deriving 
the toxicity grade.  More interesting, however, was the introduction of the MedDRA preferred term and 
system organ class associated with the toxicity.  With this, it became possible to treat laboratory toxicity 
measurements as adverse events.  When CTCAE version 5.0 was released in 2017, another layer of 
complexity was added in that the subject’s status at baseline became a factor in deriving toxicity grades.  
For some laboratory measurements, the grading is different depending on whether the subject’s baseline 
measurement was normal or abnormal.  With this addition, we cannot rely solely on ranges to derive the 
toxicity grade.  The purpose of this paper is to demonstrate how to derive toxicity grades with the new 
CTCAE 5.0 criteria. Additionally, we will discuss how to leverage the MedDRA information introduced in 
CTCAE 4.03 to represent lab toxicities as events in an ADaM OCCDS dataset.   

EVOLUTION IN COMPLEXITY  
Consider the lab parameter Alkaline Phosphatase commonly captured in the SDTM LB domain with 
LBTESTCD = “ALP”.  Abnormally high values of this parameter can result in an adverse event with the 
CTCAE term “Alkaline phosphatase increased”.  In CTCAE version 4.03, toxicity grades for this adverse 
event are calculated explicitly in ranges related to the upper limit of normal (ULN) as shown below in 
Table 1. 

Grade 1 Grade 2 Grade 3 Grade 4 

>ULN - 2.5 x ULN >2.5 - 5.0 x ULN >5.0 - 20.0 x ULN >20.0 x ULN 

Table 1: “Alkaline Phosphatase Increased” grading criteria from CTCAE version 4.03 
This is very simple to program.  With the result stored in LBSTRESN and the ULN stored in LBSTNRHI, 
one simply needs to compare the two SDTM values to determine the grade.  (Note that toxicity grade of 5 
was excluded. That grade cannot be computed based just on values in LB as it only applies when the 
event results in death.)   

Now compare this to the “Alkaline phosphatase increased” criteria for CTCAE version 5.0 as shown below 
in Table 2. 

Grade 1 Grade 2 Grade 3 Grade 4 
>ULN - 2.5 x ULN if 
baseline was normal; 2.0 - 
2.5 x baseline if baseline 
was abnormal 

>2.5 - 5.0 x ULN if baseline 
was normal; >2.5 - 5.0 x 
baseline if baseline was 
abnormal 

>5.0 - 20.0 x ULN if 
baseline was normal; >5.0 - 
20.0 x baseline if baseline 
was abnormal 

>20.0 x ULN if baseline was 
normal; >20.0 x baseline if 
baseline was abnormal 

Table 2: “Alkaline Phosphatase Increased” grading criteria from CTCAE version 5.0 
In addition to LBSTRESN and LBSTNRHI from a single record as needed with CTCAE Version 4.03, we 
also need the LBSTRESN value from the associated baseline ALP record.  First, it needs to be 
determined whether the baseline LBSTRESN is greater than the ULN.  If the baseline LBSTRESN is at or 
below the ULN, then the toxicity grade is derived based on the ULN as was done in version 4.03.  
However, if the baseline LBSTRESN is greater than the ULN, then the grades are derived by comparing 
the current LBSTRESN value to the baseline LBSTRESN value instead of the ULN.  (Note that the 
baseline records themselves are graded according to the ULN, since an abnormal baseline value cannot 
be compared to itself to get a meaningful grade.) 

DERIVING THE TOXICTY GRADES 
Deriving these more complex grade criteria can be frontloaded into an Excel spreadsheet and then called 
programmatically in a SAS macro.  The grading criteria are specified in the spreadsheet as macro 
variables that are then applied using a Call Execute routine.  The ALP increased criteria from version 5.0 
appear in the spreadsheet as shown below in Table 3. 

TESTCD TOX DIR G1 G2 

ALP 
Alkaline 
phosphatase 
increased 

(&R > &ULN OR 
&R > &B) 

((&BVISN = &VISN OR &B <= &ULN) 
AND &ULN < &R <= 2.5*&ULN) OR 
(&B > &ULN AND 2*&B <= &R <= 
2.5*&B) 

((&BVISN = &VISN OR &B <= 
&ULN) AND 2.5*&ULN < &R <= 
5*&ULN) OR (&B > &ULN AND 
2.5*&B < &R <= 5*&B) 

Table 3: “Alkaline Phosphatase Increased” grading criteria from CTCAE version 5.0 written in SAS 
macro variables. (Note that Grades 3 and 4 are excluded from this display.) 

The macro variables are defined as follows:   

• &R is the result from the record being graded, assumed to be captured in LBSTRESN.   

• &ULN is the upper limit of normal, assumed to be captured in LBSTNRHI.   

• &VISN is the visit number of the record being graded, captured in VISITNUM.   

• &B is the baseline result found in LBSTRESN of the associated baseline record.   

• &BVISN is the baseline visit number found in VISITNUM of the associated baseline record. 

The DIR column specifies a grading direction.  In this case, the direction criterion is met if the result is 
greater than the ULN or the baseline result.  While only increases are relevant for this specific test, 
specifying directionality is necessary as there are lab tests that need to have grades assigned based on 
both increases and decreases. (An example of bidirectional grading will be provided later.) 

The SAS macro for performing these derivations (see appendix for the code) starts by defining a dataset 
name, a CTCAE version, and the various macro parameters described above: 

A parameter (“ds”) refers to an SDTM domain name.  In most cases, this will be “LB”, but other SDTM 
findings domains may require toxicity grading.  The input will be an SDTM domain without the xxTOXGR 
and xxTOX variables, and the output will be the same SDTM domain with the toxicity variables added.  As 
a default, the CTCAE version is set to 5.0, but this can be manually set to other versions as needed.  Like 
the ULN parameter, the LLN parameter refers to the lower limit of normal, assumed to be collected in the 
SDTM variable LBSTNRLO.  This is not needed for the ALP test, but it is needed for low-graded CTCAE 
terms such as “Hypocalcemia”. 



 
 

3 

First, the macro determines the baseline result values.  This can be customized for specific definitions of 
baseline.  For this example, baseline is being defined as the last populated record before initial treatment.  
The SDTM IG 3.3 variable LBLOXFL (last observation before treatment flag) is used to simplify the 
derivation. Adjustments are required if an earlier SDTM IG version is being followed. 

The next step requires that the grading criteria written in macro variables as shown in Table 3 be 
available in a SAS dataset. This can be done by reading in the associated CTCAE grading criteria excel 
file each time the macro is run using proc import.  Alternatively, as in this example, the grading criteria for 
version 5.0 can be stored as a SAS dataset (here named ctcaev5).  Similarly, the version 4.03 criteria are 
stored in a SAS dataset named ctcaev4.  The version can be selected based on the macro call. The 
associated grading criteria are transposed for downstream use. 

The resulting transposed table for CTCAE version 5.0 for the “Alkaline phosphatase increased” event is 
shown below in Table 4. 
 

TESTCD TOX DIR GRADE CODE 

ALP Alkaline phosphatase 
increased 

(&R > &ULN 
OR &R > &B) G1 ((&BVISN = &VISN OR &B <= &ULN) AND &ULN < &R <= 

2.5*&ULN) OR (&B > &ULN AND 2*&B <= &R <= 2.5*&B) 

ALP Alkaline phosphatase 
increased 

(&R > &ULN 
OR &R > &B) 

G2 

((&BVISN = &VISN OR &B <= &ULN) AND &ULN < &R <= 
2.5*&ULN) OR (&B > &ULN AND 2*&B <= &R <= 2.5*&B) 

ALP Alkaline phosphatase 
increased 

(&R > &ULN 
OR &R > &B) 

G3 

((&BVISN = &VISN OR &B <= &ULN) AND 5*&ULN < &R <= 
20*&ULN) OR (&B > &ULN AND 5*&B < &R <= 20*&B) 

ALP Alkaline phosphatase 
increased 

(&R > &ULN 
OR &R > &B) 

G4 

((&BVISN = &VISN OR &B <= &ULN) AND &R > 20*&ULN) 
OR (&B > &ULN AND &R > 20*&B) 

 
Table 4: SAS dataset containing transposed grating criteria for the “Alkaline phosphatase 
increased” event 
Using this dataset, we use a Call Execute routine to run the code stored in the CODE variable (when the 
associated criteria are met). 

Lastly, a new LB dataset is created with LBTOX and LBTOXGR added, and intermediate datasets used 
for processing are dropped from the WORK library. 

With this approach, the derivations for each CTCAE and grade only need to be written out once in terms 
of macro variables. Then the Call Execute routine applies that code where needed. 
 

EXAMPLES WITH MOCK LABORATORY DATA 
Consider the mock laboratory SDTM LB dataset shown in Table 5 below: 

Row USUBJID LBSEQ LBTESTCD LBTEST LBSTRESN LBSTRESU LBSTNRLO LBSTNRHI 
1 001 1 ALP Alkaline Phosphatase 110 U/L 31 106 
2 002 1 ALP Alkaline Phosphatase 100 U/L 35 131 
3 003 1 ALP Alkaline Phosphatase 200 U/L 95 385 
4 001 2 CREAT Creatinine 0.6 mg/dL 0.4 1.1 
5 002 2 CREAT Creatinine 0.7 mg/dL 0.5 1.2 

Row USUBJID LBSEQ LBTESTCD LBTEST LBSTRESN LBSTRESU LBSTNRLO LBSTNRHI 
6 003 2 CREAT Creatinine 2.4 mg/dL 0.67 1.17 
7 001 3 CACR Calcium Corrected 2.2 mmol/L 2.15 2.55 
8 002 3 CACR Calcium Corrected 2.8 mmol/L 2.15 2.55 
9 003 3 CACR Calcium Corrected 1.6 mmol/L 2.15 2.55 

10 001 4 ALP Alkaline Phosphatase 540 U/L 31 106 
11 002 4 ALP Alkaline Phosphatase 300 U/L 35 131 
12 003 4 ALP Alkaline Phosphatase 8000 U/L 95 385 
13 001 5 CREAT Creatinine 0.84 mg/dL 0.4 1.1 
14 002 5 CREAT Creatinine 2.22 mg/dL 0.5 1.2 
15 003 5 CREAT Creatinine 7.1 mg/dL 0.67 1.17 
16 001 6 CACR Calcium Corrected 3.5 mmol/L 2.15 2.55 
17 002 6 CACR Calcium Corrected 2.03 mmol/L 2.15 2.55 
18 003 6 CACR Calcium Corrected 2.4 mmol/L 2.15 2.55 

 
Row 
(cont) LBNRIND LBLOBXFL VISITNUM VISIT LBTOX LBTOXGR 

1 HIGH Y 2 Day 1   
2 NORMAL Y 2 Day 1   
3 NORMAL Y 2 Day 1   
4 NORMAL Y 2 Day 1   
5 NORMAL Y 2 Day 1   
6 HIGH Y 2 Day 1   
7 NORMAL Y 2 Day 1   
8 HIGH Y 2 Day 1   
9 LOW Y 2 Day 1   

10 HIGH   3 Day 2   
11 HIGH   3 Day 2   
12 HIGH   3 Day 2   
13 NORMAL   3 Day 2   
14 HIGH   3 Day 2   
15 HIGH   3 Day 2   
16 HIGH   3 Day 2   
17 LOW   3 Day 2   
18 NORMAL   3 Day 2     

Table 5: Mock SDTM laboratory data used as a demonstration. (Note that some columns not shown.) 

We have data for three lab tests: Alkaline Phosphatase, Creatinine, and Calcium Corrected.  For each of 
these tests, we will examine the CTCAE derivations written in SAS macro variables for both versions 4.03 
and 5.0, and we will look at results that give a different grade between the two versions. We will show 
how to populate the LBTOX and LBTOXGR columns.   
 

ALKALINE PHOSPHATASE  
The toxicity grading criteria for the term “Alkaline phosphatase increased” for CTCAE version 5.0 is 
shown above in Table 3. The simpler toxicity grading criteria for version 4.03 is shown below in Table 6. 

TESTCD TOX DIR G1 G2 G3 G4 

ALP 
Alkaline 
phosphatase 
increased 

&R > &ULN &ULN < &R  <= 
2.5 * &ULN 

2.5 * &ULN < &R <= 
5 * &ULN 

5 * &ULN  < &R 
<= 20 * &ULN 

&R > 20 * 
&ULN 

Table 6: “Alkaline Phosphatase Increased” grading criteria from CTCAE version 4.03 written in 
SAS macro variables 
The version 4.03 criteria only compare to the ULN, while the version 5.0 criteria vary depending on 
whether the baseline assessment was normal or abnormal. 

Table 7 below shows the toxicity grades derived for subject 001 using the version 4.03 criteria, and Table 
8 shows the same records with the toxicity grades derived using the version 5.0 criteria. 

USUBJID LBTESTCD LBSTRESN LBSTNRHI LBLOBXFL VISITNUM VISIT LBTOX LBTOXGR 

001 ALP 110 106 Y 2 Day 1 
Alkaline phosphatase 
increased 1 

001 ALP 540 106   3 Day 2 
Alkaline phosphatase 
increased 3 

Table 7: Sample “Alkaline phosphatase increased” derived using CTCAE version 4.03 criteria 
 

USUBJID LBTESTCD LBSTRESN LBSTNRHI LBLOBXFL VISITNUM VISIT LBTOX LBTOXGR 

001 ALP 110 106 Y 2 Day 1 
Alkaline phosphatase 
increased 1 

001 ALP 540 106   3 Day 2 
Alkaline phosphatase 
increased 2 

Table 8: Sample “Alkaline phosphatase increased” derived using CTCAE version 5.0 criteria 
For these records, the baseline ALP measurement is 110 (as indicated by LBLOXFL), the ULN is 106, 
and the post-baseline measurement is 540.  By simply looking at 5 times the ULN (530) and 20 times the 
ULN (2120), we see that that the post-baseline measurement has a toxicity grade of 3 according to the 
version 4.03 criteria.  However, since the baseline measurement was above the ULN, we compare to 2.5 
times the baseline value (275) and 5 times the baseline value (550) and determine that the post-baseline 
measurement has a toxicity grade of 2 according to the version 5.0 criteria. 
 

CREATININE 
We will now consider the lab parameter with LBTESTCD = “CREAT”, and it’s associated CTCAE term 
“Creatinine increased”.  The version 4.03 criteria are shown below in Table 9, and the version 5.0 criteria 
are shown below in Table 10. 

TESTCD TOX DIR G1 G2 G3 G4 

CREAT Creatinine 
increased 

(&R > &ULN 
OR &R > &B) 

(&ULN < &R <= 
1.5*&ULN) OR 
(&B < &R <= 
1.5*&B) 

(1.5*&ULN < &R <= 
3*&ULN) OR 
(1.5*&B < &R <= 
3*&B) 

(3*&ULN < &R <= 
6*&ULN) OR (&R > 
3*&B) 

&R > 
6*&ULN 

Table 9: CTCAE Version 4.03 toxicity grading criteria for “Creatinine increased” written in terms of 
SAS macro variables 
 

TESTCD TOX DIR G1 G2 G3 G4 

CREAT Creatinine 
increased 

(&R > 
&ULN OR 
&R > &B) 

&ULN < &R <= 
1.5*&ULN 

(1.5*&ULN < &R <= 
3*&ULN) OR 
(1.5*&B < &R <= 
3*&B) 

(3*&ULN < &R <= 
6*&ULN) OR (&R > 
3*&B) 

&R > 6*&ULN 

Table 10: CTCAE Version 5.0 toxicity grading criteria for “Creatinine increased” written in terms of 
SAS macro variables 
While there is no major change in complexity between version 4.03 and version 5.0 in this example 
(unlike with “Alkaline phosphatase increased”) there is a subtle difference in the way a toxicity grade of 1 
is derived that can cause errors if care is not taken when applying the different CTCAE versions.  In 
version 4.03, a record with a result that is greater than the baseline value but not greater than the ULN 
can be given a toxicity grade of 1.  This relationship was removed in version 5.0 so that records in the 
normal range are not given a toxicity grade unless they are greater than 1.5 times the baseline value. 

Table 11 below shows the toxicity grades derived for subject 001 using the version 4.03 criteria, and 
Table 12 below shows the same records with the toxicity grades derived using the version 5.0 criteria. 

USUBJID LBTESTCD LBSTRESN LBSTNRHI LBLOBXFL VISITNUM VISIT LBTOX LBTOXGR 
001 CREAT 0.6 1.1 Y 2 Day 1     
001 CREAT 0.84 1.1   3 Day 2 Creatinine increased 1 

Table 11: Sample “Creatinine increased” derived using CTCAE version 4.03 criteria 
 

USUBJID LBTESTCD LBSTRESN LBSTNRHI LBLOBXFL VISITNUM VISIT LBTOX LBTOXGR 
001 CREAT 0.6 1.1 Y 2 Day 1     
001 CREAT 0.84 1.1   3 Day 2   

Table 12: Sample “Creatinine increased” derived using CTCAE version 5.0 criteria 
 

In this example, the post-baseline measurement of 0.84 is below the ULN of 1.1.  However, since the 
result is greater than the baseline result of 0.6, the post-baseline measurement is given a toxicity grade of 
1 using the version 4.03 criteria.  With the version 5 criteria, the post-baseline measurement is still less 
than 1.5 times the baseline measurement (0.6*1.5=0.9), so the record is not given a toxicity grade at all.  
When transitioning from version 4.03 to version 5.0, failure to notice subtle changes like this can result in 
unnecessary toxicity events appearing in the study data. 

CALCIUM 
While calcium toxicity grading is simple and identical between CTCAE versions 4.03 and 5.0, this 
example is included to show how this method handles lab tests that can be graded in two directions.  
Calcium has two associated CTCAE terms; “Hypocalcemia” and “Hypercalcemia”.  The CTCAE criteria for 
these terms are shown in Table 13 and Table 14 below. 

 
Table 13: Hypocalcemia toxicity grading criteria (versions 4.03 and 5.0)  
 

 
Table 14: Hypercalcemia toxicity grading criteria *Versions 4.03 and 5.0) 
Please note that this technique does not handle descriptive terms such as “symptomatic”, “hospitalization 
indicated”, “life-threatening consequences”, or “Death”, since these cannot be determined from lab data 
alone.  Also notice that there are multiple lab tests and units that can lead to a CTCAE term of 
“Hypocalcemia” or “Hypercalcemia”.  The mock data in this example assumes that corrected serum 
calcium (LBTESTCD = “CACR”) is collected with a standardized unit of LBSTRESU = “mmol/L”. 

These grading criteria in both versions 4.03 and 5.0, written in terms of SAS macro variables, are shown 
in Table 15 below. 

TESTCD TOX DIR G1 G2 G3 G4 

CACR Hypercalcemia &R > &ULN &ULN < &R <= 2.9 2.9 < &R <= 3.1 3.1 < &R <= 3.4 &R > 3.4 

CACR Hypocalcemia &R < &LLN 2 <= &R < &LLN 1.75 <= &R < 2 1.5 <= &R < 
1.75 &R < 1.5 

Table 15: CTCAE grading criteria for “Hypocalcemia” and “Hypercalcemia” (Versions 4.03 and 5.0) 
This scenario shows the utility for the DIR column.  By checking the result against the ULN and the LLN in 
the beginning of the Call Execute portion of the macro, we can ensure that toxicity grades are only 
derived for the appropriate LBTOX value.  For reference, the CACR records from the mock LB dataset 
with LBTOXGR and LBTOX derived is included below in Table 16. 

USUBJID LBSEQ LBTESTCD LBSTRESN LBSTNRLO LBSTNRHI LBNRIND LBLOBXFL VISIT LBTOX LBTOXGR 
001 3 CACR 2.2 2.15 2.55 NORMAL Y Day 1     
001 6 CACR 3.5 2.15 2.55 HIGH   Day 2 Hypercalcemia 4 
002 3 CACR 2.8 2.15 2.55 HIGH Y Day 1 Hypercalcemia 1 
002 6 CACR 2.03 2.15 2.55 LOW   Day 2 Hypocalcemia 1 
003 3 CACR 1.6 2.15 2.55 LOW Y Day 1 Hypocalcemia 3 
003 6 CACR 2.4 2.15 2.55 NORMAL   Day 2     

Table 16: Corrected Calcium data with toxicity grades derived 

LAB TOXICITIES AS EVENTS IN AN OCCDS DATASET 
With the addition of MedDRA terminology into the CTCAE criteria, it is now possible to treat lab toxicities 
as events in an OCCDS dataset.  Consider the output below in Table 17 

 
Table 17: Laboratory toxicities analyzed as events using the MedDRA System Organ Class and 
Preferred Term (Minjoe and Widel, 2013) 
With the creation of the OCCDS standard, there is now a formal compliant and official format for 
supporting analysis data for events other than AEs.  With the MedDRA terms available in LBTOX and an 

official standard available for occurrence data, it is now possible to summarize lab toxicities as events as 
shown in Table 17 in a compliant manner.  Sample records from the associated ADaM dataset, which we 
call ADLBTOX, are shown below in Table 18. 

USUBJID LBTESTCD LBSEQ LBTOX ASOC LBTOXGR TRTP 
001 ALB 3 Hypoalbuminemia Blood and Lymphatic System Disorders 2 1 
002 ALB 6 Hypoalbuminemia Blood and Lymphatic System Disorders 3 1 
003 ALB 7 Hypoalbuminemia Blood and Lymphatic System Disorders 4 2 

Table 18: Sample records from an OCCDS ADLBTOX dataset that can support analysis of lab 
toxicity events 
Only selected columns are shown in order to support the analysis in Table 17.  The SDTM variables 
LBTESTCD and LBSEQ are included to maintain traceability from LB to ADLBTOX. 

CONCLUSION 
Even with the additional complexity added to the definitions of laboratory toxicity grades that comes from 
considering baseline values, deriving these grades can still be handled easily and consistently with proper 
planning. By housing all the grade derivations into a reference metadata as macro variables and utilizing 
the Call Execute routine in SAS, these grades (regardless of CTCAE version) can be derived 
automatically and efficiently.  Additionally, since this process requires that all toxicity grade definitions be 
stored in a single file, that file can be maintained (or at least easily reviewed) by an expert in laboratory 
assessments. 

APPENDIX 
%macro tox(ds,version=5,R=lbstresn,B=base,ULN=lbstnrhi,LLN=lbstnrlo, 
            VISN=visitnum,BVISN=basevisit); 
 
/*get baseline values in the same row*/ 
proc sql; 
create table base as 

select usubjid, &ds.testcd, &ds.stresn as base, visitnum as  basevisit 
  from &ds. 
   where &ds.lobxfl = 'Y' 
; 
 
 
/*join back to the dataset*/ 
create table &ds._base as 
 select a.*, b.base, b.basevisit 
  from &ds. a 
   left join base b 
    on a.usubjid = b.usubjid 
     and a.&ds.testcd = b.&ds.testcd 
; 

quit; 
/*sort CTCAE dataset for transpose*/ 
proc sort data = %if &version = 4 %then %str(ctcaev4); %else %if &version = 
5 %then %str(ctcaev5);; 
 by testcd tox dir; 
run; 
 
/*transpose to get grading in a single column*/ 
proc transpose data =  
 %if &version = 4 %then %str(ctcaev4); %else %if &version = 5 %then 
%str(ctcaev5); 
 out = ctcae (rename = (_NAME_ = GRADE COL1 = CODE) drop = _LABEL_); 
 by testcd tox dir; 
 var g:; 
run;  
/*Data step that executes another data step by reading in code form the 
CTCAE dataset*/ 
data _null_; 
 set ctcae end=eof; 
 if _n_ = 1 then  
  do; 
   call execute('data &ds._tox;'); 
   call execute(' set &ds._base;'); 
  end; 
 

call execute('if &ds.testcd = "'||strip(testcd)||'" and 
'||strip(dir)||' then do;'); 

call execute(' if '||strip(code)||' then &ds.TOXGR = 
"'||substr(strip(grade),2)||'";'); 

call execute(' if lbtoxgr ne "" then &ds.TOX = 
"'||strip(tox)||'";'); 
  call execute('end;'); 
 
 if eof then 
  do; 

   call execute('run;'); 
  end; 

run; 
 

/*Create new dataset with toxicity variables added*/ 
data &ds.; 
 set &ds._tox (drop = base basevisit); 
run; 
 
/*delete intermediate processing datasets*/ 
proc datasets nolist; 
delete base ctcae lb_base lb_tox; 
 
%mend; 

